Math 2650

A. J. Meir


```
> restart:with(DEtools):with(plots):
```

Euler's method

Consider the equation

$$\frac{dy}{dx} = y^2 \qquad y(0) = 1$$

We verify that

```
> y:= x-> 1/(1-x);
```

$$y := x \to \frac{1}{1-x} \tag{1.1}$$

is a solution. On the interval (0, 1), it is a differenitiable function, and

```
> D(y)(x)-y(x)^2;
```

$$0 \tag{1.2}$$

so it satisfies the equation, and

```
> y(0);
```

$$1 \tag{1.3}$$

so it satisfies the initial condition, hence it is a solution.

Lets try and use Euler's method to construct an approximate solution

Below we write a little Maple procedure for the forward Euler's method.

Euler's Method

```
> FEuler:=proc(f,t0,x0,h,N)
  local t,x,n,sol:
  t:=evalf(t0):
  x:=evalf(x0):
  sol:=[[t,x]]:
  for n from 1 to N do
  x:=x+h*f(t,x):
  t:=t+h:
  sol:=[op(sol),[t,x]]:
  od;
  end;
```

FEuler := **proc**(*f*, *t0*, *x0*, *h*, *N*)

local *t*, *x*, *n*, *sol*;

```
    t := evalf(t0);
    x := evalf(x0);
    sol := [[t, x]];
    for n to N do
        x := x + h*f(t, x); t := t + h; sol := [op(sol), [t, x]]
    end do
end proc
```

```
> y:='y';
```

$$y := y \qquad (1.4)$$

```
> x0:=0;xf:=1.2;y0:=1;
```

$$x0 := 0$$

$$xf := 1.2$$

$$y0 := 1$$

```
> N:=6;
```

$$N := 6$$

```
> h:=evalf((xf-x0)/N);
```

$$h := 0.2000000000$$

```
> f:=(x,y)->y^2;
```

$$f := (x, y) \rightarrow y^2$$

```
> out:=FEuler(f,x0,y0,h,N);
```

$out := [[0., 1.], [0.2000000000, 1.200000000], [0.4000000000, 1.488000000], [0.6000000000, 1.930828800], [0.8000000000, 2.676448771], [1.000000000, 4.109124376], [1.200000000, 7.486105004]]$

```
> EulerApprox:=plot(out,color=green,thickness=2):
> DF:=DEplot(diff(y(x),x)=y(x)^2,y(x),x=-1..1.5,y=0..8):
> EX:=plot(1/(1-x),x=0..1,color=blue,thickness=2):
> display(EulerApprox,DF,EX);
```



Note the exact solution (blue) has a vertical asymptote at x=1. The approximate solution (green) misses this feature,

and suggests that the solution is valid for values of x greater than 1. Note also the mismatch between the slope lines
and the approximate solution. This is of course wrong, and is due to the finite step size.

Now for

$$\frac{dy}{dx} = x^2 + y \quad y(0) = 1$$

we will use Euler's method again. First with a stepsize h=0.1

```
> x0:=0;xf:=1;y0:=1;
```

$$x0 := 0$$

$$xf := 1$$

$$y0 := 1$$

```
> N:=10;
```

$$N := 10$$

```
> h:=evalf((xf-x0)/N);
```

$$h := 0.1000000000$$

```
> f:=(x,y)->x^2+y;
```

$$f := (x, y) \rightarrow x^2 + y$$

```
> out01:=FEuler(f,x0,y0,h,N);
```

out01 := [[0., 1.], [0.1000000000, 1.100000000], [0.2000000000, 1.211000000], [0.3000000000, 1.336100000], [0.4000000000, 1.478710000], [0.5000000000, 1.642581000], [0.6000000000, 1.831839100], [0.7000000000, 2.051023010], [0.8000000000, 2.305125311], [0.9000000000, 2.599637842], [1.000000000, 2.940601626]]

Next with a stepsize h=0.05

```
> N:=20;
```

$$N := 20$$

```
> h:=evalf((xf-x0)/N);
```

$$h := 0.05000000000$$

```
> out005:=FEuler(f,x0,y0,h,N);
```

out005 := [[0., 1.], [0.05000000000, 1.050000000], [0.1000000000, 1.102625000], [0.1500000000, 1.158256250], [0.2000000000, 1.217294062], [0.2500000000, 1.280158765], [0.3000000000, 1.347291703], [0.3500000000, 1.419156288], [0.4000000000, 1.496239102], [0.4500000000, 1.579051057], [0.5000000000, 1.668128610], [0.5500000000, 1.764035040], [0.6000000000, 1.867361792], [0.6500000000, 1.978729882], [0.7000000000, 2.098791376], [0.7500000000, 2.228230945], [0.8000000000, 2.367767492], [0.8500000000, 2.518155867], [0.9000000000, 2.680188660], [0.9500000000, 2.854698093], [1.000000000, 3.042557998]] **(1.5)**

```
> O1:=plot(out01,color=green,thickness=2):
> O05:=plot(out005,color=blue,thickness=2):
> display(O1,O05);
```



The accuracy of the approximate solution (the numerical solution) increases as the step size decreases.